
Forward Proof in HOL

Forward Proof in HOL

x The so-called forward proof style:

` thm

�
�

�
�RULE

P
P

P
P

P
P

P
PP

�
�

�
�

�
�

�
��

` thm` thm

�
�

�
�RULE

P
P

P
P

P
P

P
PP

�
�

�
�

�
�

�
��

` thm ` thm` thm

�
�

�
�RULE

` thm

desired:

known:

x Forward proofs:

u can be millions of (primitive) inferences long

u usually not natural for ‘one-off’ proofs

u but essential for tool building

Forward Proof and Primitive Rules

x All theorems in HOL are ultimately proved
using only the primitive inference rules:

SUBST:
Γ1 ` N1 = M1 . . . Γn ` Nn = Mn Γ ` P

Γ1 ∪ . . . ∪ Γn ∪ Γ ` P[M1, . . . , Mn/N1, . . . , Nn]

DISCH:
Γ ` P

Γ − {Q} ` Q ⊃ P
MP:

Γ1 ` P ⊃ Q Γ2 ` P

Γ1 ∪ Γ2 ` Q

INST_TYPE:
Γ ` P

Γ ` P[σ1, . . . , σn/α1, . . . , αn]

ABS:
Γ ` M = N

Γ ` (λv. M) = (λv. N)
(v not free in Γ)

BETA_CONV:
` (λv. N) M = N[M/v]

REFL:
` M = M

ASSUME:
{P} ` P

x We shall now look at the ML representations
of these rules.

Introducing an Assumption

x In logic, the rule is:

ASSUME:
{P} ` P

x This rule allows us to deduce {P} ` P for any
boolean term P, so the HOL rule

ASSUME : term -> thm

takes this term as an argument.

Examples

x Examples using ASSUME:

- ASSUME;
> val it = fn : term -> thm

- ASSUME ‘‘Q /\ R‘‘;
> val it = . |- Q /\ R : thm

- show_assums := true;
> val it = () : unit

- ASSUME ‘‘Q /\ R‘‘;
> val it = [Q /\ R] |- Q /\ R : thm

- ASSUME ‘‘n:num‘‘ handle e => Raise e;
Exception raised at Thm.ASSUME:
not a proposition

Reflexivity of Equality

x The rule is:

REFL:
` M = M

x The HOL function takes a term argument:

- REFL ‘‘T‘‘;
> val it = |- T = T : thm

- REFL ‘‘x:num‘‘;
> val it = |- x = x : thm

Discharging an Assumption

x The rule is:

DISCH:
Γ ` P

Γ − {Q} ` Q ⊃ P

x Example:

- val th = ASSUME ‘‘F‘‘;
> val th = . |- F : thm

- DISCH;
> val it = fn : term -> thm -> thm

- DISCH ‘‘F‘‘ th;
> val it = |- F ==> F : thm

- DISCH ‘‘T‘‘ th;
> val it = . |- T ==> F : thm

- DISCH ‘‘x=3‘‘ (REFL ‘‘x=9‘‘);
> val it = |- (x = 3) ==> ((x = 9) = x = 9) : thm

Modus Ponens

x The rule is:

MP:
Γ1 ` P ⊃ Q Γ2 ` P

Γ1 ∪ Γ2 ` Q

x Implementation in HOL:

- MP;
> val it = fn : thm -> thm -> thm

- val th1 = DISCH ‘‘T=T‘‘ (REFL ‘‘F‘‘);
> val th1 = |- (T = T) ==> (F = F) : thm

- MP th1 (REFL ‘‘T‘‘);
> val it = |- F = F : thm

x Modus ponens works up to α-equivalence:

- val th1 =
ASSUME ‘‘(!x:num. Q x) ==> ?x. Q x‘‘;

> val th1 = . |- (!x. Q x) ==> (?x. Q x) : thm

- val th2 = ASSUME ‘‘!y:num. Q y‘‘;
> val th2 = . |- !y. Q y : thm

- MP th1 th2;
> val it = .. |- ?x. Q x : thm

The Abstraction Rule

x The rule is:

ABS:
Γ ` M = N

Γ ` (λv. M) = (λv. N)

where the variable v is not free in Γ.

x An example:

- ABS;
> val it = fn : term -> thm -> thm

- val th = REFL ‘‘x:num‘‘;
> val th = |- x = x : thm

- ABS ‘‘x:num‘‘ th;
> val it = |- (\x. x) = (\x. x) : thm

- ABS ‘‘y:’a‘‘ it;
> val it = |- (\y x. x) = (\y x. x) : thm

Abstraction: Failure Conditions

x The term argument must be a variable:

- ABS ‘‘x+y‘‘ (ASSUME ‘‘(x:’a) = x‘‘)
handle e => Raise e;

Exception raised at Thm.ABS:
first argument is not a variable
! Uncaught exception ....

x The theorem must be an equation:

- ABS ‘‘x:’a‘‘ (ASSUME ‘‘T‘‘)
handle e => Raise e;

Exception raised at Thm.ABS:
not an equality
! Uncaught exception ....

x The variable must not be free in the
assumptions of the theorem:

- val th = ASSUME ‘‘(x:’a) = y‘‘;
> val th = . |- x = y : thm

- ABS ‘‘x:’a‘‘ th
handle e => Raise e;

Exception raised at Thm.ABS:
The variable is free in the assumptions
! Uncaught exception ....

Beta Conversion

x The β-conversion rule is:

BETA_CONV:
` (λv. N) M = N[M/v]

x In HOL, the function BETA_CONV takes the term
(λv. N) M as an argument.

x Example:

- BETA_CONV;
> val it = fn : term -> thm

- BETA_CONV ‘‘(\x. x=3) 4‘‘;
> val it = |- (\x. x = 3) 4 = 4 = 3 : thm

- BETA_CONV ‘‘(\x:’a. \y. x=y) y‘‘;
> val it = |- (\x y. x = y) y = (\y’. y = y’) : thm

- BETA_CONV ‘‘x=0‘‘ handle e => Raise e;
Exception raised at Thm.BETA_CONV:
not a beta-redex

Substitution

x The rule for substitution is SUBST:

Γ1 ` N1 = M1 . . . Γn ` Nn = Mn Γ ` P

Γ1 ∪ . . . ∪ Γn ∪ Γ ` P[M1, . . . , Mn/N1, . . . , Nn]

where

P[M1, . . . , Mn/N1, . . . , Nn]

means P with designated free occurrences of Ni

replaced by Mi.

x The ‘designating’ of occurrences to replace
turns out to be a bit complicated:

- SUBST;
> val it = fn : {redex: term, residue: thm} list

-> term -> thm -> thm

Substitution (continued)

x SUBST takes three arguments:

(thm * term) list
︸ ︷︷ ︸

a list of pairs of substitution equa-

tions and marker variables of the

form ( ` N1 = Mi, v)

-> term
︸ ︷︷ ︸

a template with

marked subterms

-> thm
︸ ︷︷ ︸

Γ2 ` P

-> thm

x A simple example:

- val th1 = ASSUME ‘‘x=7‘‘;
> val th1 = . |- x = 7 : thm

- SUBST [{redex = ‘‘m:num‘‘, residue = th1}]
‘‘m+x = x+m‘‘ (REFL ‘‘x+x‘‘);

> val it = . |- 7 + x = x + 7 : thm

- SUBST [{redex = ‘‘m:num‘‘, residue = th1}]
‘‘x+x = m+x‘‘ (REFL ‘‘x+x‘‘);

> val it = . |- x + x = 7 + x : thm

More Substitution Examples

x Another example:

- val th1 = BETA_CONV ‘‘(\x. x) 1‘‘;
> val th1 = |- (\x. x) 1 = 1 : thm

- val th2 = REFL ‘‘(\x.x) 1‘‘;
> val th2 = |- (\x. x) 1 = (\x. x) 1 : thm

- val th3 = SUBST [{redex ‘‘m:num‘‘,residue=th1}]
‘‘m = (\x.x) 1‘‘ th2;

> val th3 = |- 1 = (\x. x) 1 : thm

- SUBST [redex ‘‘m:num‘‘,residue=th3}]
- ‘‘1 = (\x.x)m‘‘ th3;
> val it = |- 1 = (\x. x) ((\x. x) 1) : thm

x SUBST allows simultaneous substitutions:
- val th1 = ASSUME ‘‘x=7‘‘
and th2 = ASSUME ‘‘y=x+10‘‘;

> val th1 = . |- x = 7 : thm
> val th2 = . |- y = x + 10 : thm

- val th3 = REFL ‘‘x+y‘‘;
> val th3 = |- x + y = x + y : thm

- SUBST [{redex ‘‘m1:num‘‘, residue= th1},
{redex ‘‘m2:num‘‘, residue= th2}]
‘‘m1+y = x+m2‘‘ th3;

> val it = .. |- 7 + y = x + x + 10 : thm

Type Instantiation

x The type instantiation rule is:

INST_TYPE:
Γ ` P

Γ ` P[σ1, . . . , σn/α1, . . . , αn]

x The first argument to INST_TYPE is a list of the
type instantiations to be made:

- INST_TYPE;
> val it = fn:(hol_type*hol_type)list -> thm -> thm

- show_types := true;
> val it = () : unit

- val th = REFL ‘‘\x:’a. f(x) = (y:’b)‘‘;
> val th =
|- (\(x :’a). (f :’a -> ’b) x = (y :’b)) =

(\(x :’a). (f :’a -> ’b) x = (y :’b)) :thm

- INST_TYPE [(‘‘:num‘‘, ‘‘:’a‘‘),
(‘‘:bool‘‘, ‘‘:’b‘‘)] th;

> val it =
|- (\(x:num). (f:num -> bool) x = (y:bool)) =

(\(x:num). (f:num -> bool) x = (y:bool)) :thm

Example of Forward Proof

x The proof of ` ¬F we did before:

1) ` (λb. b ⊃ F) F = (F ⊃ F) BETA_CONV

2) ` (λb. b ⊃ F) F = (λb. b ⊃ F) F REFL

3) ` (F ⊃ F) = (λb. b ⊃ F) F SUBST 1,2

4) {F} ` F ASSUME

5) ` F ⊃ F DISCH 4

6) ` (λb. b ⊃ F) F SUBST 3,5

7) ` ¬ = (λb. b ⊃ F) DEFN of ¬

8) ` ¬ = ¬ REFL

9) ` (λb. b ⊃ F) = ¬ SUBST 7,8

10) ` ¬F SUBST 6,9

x This proof uses only primitive inference rules.

HOL/FM: S-8 c© T F Melham 1993 Forward Proof: 16



The Proof in HOL

x Apply beta-conversion:

- val t1 = BETA_CONV ‘‘(\b. b ==> F) F‘‘;
> val t1 = |- (\b. b ==> F) F = F ==> F : thm

x Reverse the equation just derived:

- val t2 = REFL ‘‘(\b. b==> F) F‘‘;
> val t2 = |- (\b. b ==> F) F = (\b. b ==> F) F :thm

- val t3 = SUBST [{redex ‘‘m:bool‘‘,residue=t1}]
‘‘m = (\b. b==>F) F‘‘ t2;

> val t3 = |- F ==> F = (\b. b ==> F) F : thm

x Prove |- F ==> F using ASSUME and DISCH:

- val t4 = ASSUME ‘‘F‘‘;
> val t4 = . |- F : thm

- val t5 = DISCH ‘‘F‘‘ t4;
> val t5 = |- F ==> F : thm

x Conclude as follows:

- val t6 = SUBST [{redex ‘‘m:bool‘‘,residue=t3}]
‘‘m:bool‘‘ t5;

> val t6 = |- (\b. b ==> F) F : thm

The Proof Concluded

x The theorem just derived:

- t6;
> val it = |- (\b. b ==> F) F : thm

x Reverse the defining equation for negation:

- val t7 = REFL ‘‘$~‘‘;
> val t7 = |- ~ = ~ : thm

- val t8 = SUBST [{redex = ‘‘m:bool->bool‘‘,
residue = NOT_DEF}]
‘‘m = $~‘‘ t7;

> val t8 = |- (\t. t ==> F) = ~ : thm

x Substitute with this equation:

- val th = SUBST
[{redex ‘‘m:bool->bool‘‘,residue=t8}]
‘‘m F:bool‘‘ t6;

> val th = |- ~F : thm

HOL/FM: S-8 c© T F Melham 1993 Forward Proof: 18

Derived Inference Rules

x Derived inference rules are just ML programs.

x The rule we looked at before:

SYM:
Γ ` M = N

Γ ` N = M

1) Γ ` M = N HYPOTHESIS

2) ` M = M REFL

3) Γ ` N = M SUBST 1,2

And the ML code that implements it:

fun SYM th =
let val (left,right) = dest_eq(concl th)

val v = genvar(type_of left)
in
SUBST [{redex=v, residue=th}]

(mk_eq(v,left)) (REFL left)
end handle _ => failwith "SYM: not an equation";

x Let’s look at how one develops this code. . .

Deriving the Rule in HOL

x Start by assuming a typical hypothesis:

- val th = ASSUME ‘‘(M:’a) = N‘‘;
> val th = . |- M = N : thm

x We need to know what ‘M’ and ‘N’ are:

- val (left,right) = dest_eq(concl th);
> val left = ‘‘M‘‘ : term
> val right = ‘‘N‘‘ : term

x We need a marker variable:

- val v = genvar(type_of left);
> val v = ‘‘ %%genvar%%360 ‘‘ : term

x The result is then given by:

- SUBST [{redex=v, residue=th}]
> ‘‘ ^v = ^left ‘‘ (REFL left);
> val it = . |- N = M : thm

The Derived Rule

x Since M and N were arbitrary, this derivation
suggests the following code:

- fun SYM th =
let val (left,right) = dest_eq(concl th)

val v = genvar(type_of left)
in
SUBST [{redex=v, residue=th}]

(mk_eq(v,left)) (REFL left)
end handle _ => failwith "SYM: not an equation";

> val SYM = fn : thm -> thm

x Test the rule:

- F_DEF;
> val it = |- F = (!t. t) : thm

- SYM (F_DEF);
> val it = |- (!t. t) = F : thm

- val thm = ASSUME ‘‘!b. b = T‘‘;
> val thm = . |- !b. b = T : thm

- SYM thm handle e => Raise e;
Exception raised at ??.failwith:
SYM: not an equation
! Uncaught exception: ....

Built-in Derived Rules

x There is a wide range of derived inference rules
built into the system:

-

increasing complexity and power

primitive rules

derived logical rules

rewriting inference rules

boolean decision procedures

arithmetic decision procedures

automatic type definitions

x To become an expert HOL user, you should aim
always to be learning new rules and proof tech-
niques.

Example of Forward Proof

x We will use forward proof to derive

((∃x. R x) ⊃ Q) = (∀x. R x ⊃ Q)

by proving implication in both directions.

x Show hypotheses during the proof:

- show_assums := true;
> val it = () : unit

x To prove the left-to-right implication, first
assume the antecedant:

- val th1 = ASSUME ‘‘(?x:’a. R(x))==> Q‘‘;
> val th1 = [(?x. R x)==> Q] |- (?x. R x)==> Q :thm

x Now, assume R x:

- val th2 = ASSUME ‘‘R(x:’a):bool‘‘;
> val th2 = R x |- R x

Example Continued

x We can now use the rule

Γ ` P[u/x]

Γ ` ∃x. P

to deduce ?x. R x:

- th2;
> val it = [R x] |- R x : thm

- val th3 = EXISTS ‘‘?x:’a. R(x)‘‘,
‘‘x:’a‘‘ th2;

> val th3 = [R x] |- ?x. R x : thm

x Now, infer Q:

- th1;
> val it = [(?x. R x) ==> Q] |- (?x. R x) ==> Q :thm

- val th4 = MP th1 th3;
> val th4 = [(?x. R x) ==> Q, R x] |- Q : thm

x Discharge the assumption R x:

- val th5 = DISCH ‘‘R(x:’a):bool‘‘ th4;
> val th5 = [(?x. R x) ==> Q] |- R x ==> Q : thm

Example Continued

x Generalize x and discharge the assumption:

- val th6 = DISCH_ALL (GEN ‘‘x:’a‘‘ th5);
> val th6 = |- ((?x. R x) ==> Q) ==>

(!x. R x ==> Q) : thm

x To prove the reverse implication, again start
by assuming the antecedent:

- val th7 = ASSUME ‘‘!(x:’a).R(x) ==> Q‘‘;
> val th7 = [!x. R x ==> Q] |- !x. R x ==> Q : thm

x Specialize to x:

- th7;
> val it = [!x. R x ==> Q] |- !x. R x ==> Q : thm

- val th8 = SPEC ‘‘x:’a‘‘ th7;
> val th8 = [!x. R x ==> Q] |- R x ==> Q : thm

x Move R x into the assumptions:

- val th9 = UNDISCH th8;
> val th9 = [!x. R x ==> Q, R x] |- Q : thm

Example Continued

x Now, assume ?x. R x:

- val th10 = ASSUME ‘‘?x:’a. R(x)‘‘;
> val th10 = [?x. R x] |- ?x. R x :thm

x We can now use the rule

Γ1 ` ∃x. R Γ2 ∪ {R[v/x]} ` Q

Γ1 ∪ Γ2 ` Q
CHOOSE ("v",(Γ1 ` ∃x.R))

to conclude Q:

- th9;
> val it = [!x. R x ==> Q, R x] |- Q :thm

- val th11 = CHOOSE (‘‘x:’a‘‘, th10) th9;
> val th11 = [?x. R x, !x. R x ==> Q] |- Q :thm

x Discharge the assumption ?x. R x:

- val th12 = DISCH ‘‘?x:’a. R(x)‘‘ th11;
> val th12 = [!x. R x ==> Q] |- (?x. R x) ==> Q :thm

The Example Concluded

x The results so far:

- th6;
> val it = |- ((?x. R x) ==> Q) ==>

(!x. R x ==> Q) :thm

- th12;
> val it = [!x. R x ==> Q] |- (?x. R x) ==> Q :thm

x Discharge the remaining assumption:

- val th13 = DISCH_ALL th12;
> val th13 = |- (!x. R x ==> Q) ==>

(?x. R x) ==> Q :thm

x Put the two implications together:

- val thm = IMP_ANTISYM_RULE th6 th13;
> val thm = |- (?x. R x)==> Q = (!x. R x ==> Q) :thm

x Q.E.D.

Summary — Forward Proof

x Primitive Inference Rules:

ASSUME, REFL, DISCH, MP, ABS,

BETA_CONV, SUBST, INST_TYPE.

x Derived Inference Rules:

SYM, EXISTS, DISCH_ALL, SPEC, GEN,

UNDISCH, CHOOSE, IMP_ANTISYM_RULE.

There are many more built-in derived rules.

Rewriting Inference Rules

x A powerful set of built-in inference rules is
provided for rewriting.

x Rewriting is different from substitution:

u it deals with pattern matching

u it deals with bound variables

x Example:

- val num_CASES = arithmeticTheory.num_CASES;
> val num_CASES = |- !m.(m=0) \/ (?n. m = SUC n):thm

- val ADD1 = arithmeticTheory.ADD1;
> val ADD1 = |- !m. SUC m = m + 1 : thm

- REWRITE_RULE [ADD1] num_CASES;
> val it = |- !m. (m = 0) \/ (?n. m = n + 1) : thm

x The rewriting rule automatically:

u specialized m to n in ADD1

u substutituted under the bound variable n

Basic Rewriting Rules

x There are two basic rules:

PURE_REWRITE_RULE : thm list -> thm -> thm

REWRITE_RULE : thm list
︸ ︷︷ ︸

list of equational theorems to be
used as left-to-right rewrite rules

− > thm
︸ ︷︷ ︸

theorem to be rewritten

-> thm

x Rewriting is done:

u with all the supplied equations

u on all subterms of the theorem to be rewritten

u repeatedly, until no rewrite rule applies

x In addition to equations, rewrite rules can be:

u quantified: |- !x1. . . xn. LHS = RHS

u negations, which become |- LHS = F

u non-equations, which become |- LHS = T

u conjunctions of the above

Basic Rewrite Rules

x PURE_REWRITE_RULE rewrites with only the
supplied list of theorems:

- val th =
ASSUME ‘‘!x:num.(P /\ (SUC 0=1)) ==> F‘‘;

> val th = . |- !x. P /\ (SUC 0 = 1) ==> F : thm

- val one = num_CONV ‘‘1‘‘;
> val one = |- 1 = SUC 0 : thm

- PURE_REWRITE_RULE [one] th;
> val it = . |- !x. P /\ (SUC 0 = SUC 0) ==> F : thm

x REWRITE_RULE also rewrites with some standard
simplifications:

- REWRITE_RULE [one] th;
> val it = . |- ~P : thm

Standard Simplifications

x REWRITE_RULE uses these simplifications:

|- !x. (x = x) = T
|- !t.

((T = t) = t) /\ ((t = T) = t) /\
((F = t) = ~t) /\ ((t = F) = ~t)

|- (!t. ~~t = t) /\ (~T = F) /\ (~F = T)
|- !t. (T /\ t = t) /\ (t /\ T = t) /\

(F /\ t = F) /\ (t /\ F = F) /\
(t /\ t = t)

|- !t. (T \/ t = T) /\ (t \/ T = T) /\
(F \/ t = t) /\ (t \/ F = t) /\
(t \/ t = t)

|- !t. (T ==> t = t) /\ (t ==> T = T) /\
(F ==> t = T) /\ (t ==> t = T) /\
(t ==> F = ~t)

|- !t1 t2. ((T => t1 | t2) = t1) /\
((F => t1 | t2) = t2)

|- !t. (!x. t) = t
|- !t. (?x. t) = t
|- !t1 t2. (\x. t1)t2 = t1
|- !x. FST x,SND x = x
|- !x y. FST(x,y) = x
|- !x y. SND(x,y) = y

x If you do not need these, PURE_REWRITE_RULE is
often much faster.

Other Rewriting Rules

x Rewriting using the assumptions:

ASM_REWRITE_RULE

ASM_PURE_REWRITE_RULE

These add the assumptions of the theorem to
be rewritten to the list of rewrites.

x Rewriting only once:

ONCE_REWRITE_RULE

ONCE_ASM_REWRITE_RULE

PURE_ONCE_REWRITE_RULE

PURE_ONCE_ASM_REWRITE_RULE

These apply at most one rewrite rule to any
subterm.

Summary of Rewriting

x Rewrite using standard simpifications:

u REWRITE_RULE

u ASM_REWRITE_RULE

u ONCE_REWRITE_RULE

u ONCE_ASM_REWRITE_RULE

x Rewrite without standard simpifications:

u PURE_REWRITE_RULE

u PURE_ASM_REWRITE_RULE

u PURE_ONCE_REWRITE_RULE

u PURE_ONCE_ASM_REWRITE_RULE

x Rewrite using assumptions:

u ASM_REWRITE_RULE

u PURE_ASM_REWRITE_RULE

u ONCE_ASM_REWRITE_RULE

u PURE_ONCE_ASM_REWRITE_RULE

